Eigenvalues and Eigenvectors

```
> restart;
> with(LinearAlgebra):
```

We illustrate the process of finding **eigenvalues** and **eigenvectors** with a 3x3 matrix.

```
> A:=Matrix(3,3,[[0,2,-1],[1,-5,3],[2,0,1]]);
```

$$A := \begin{bmatrix} 0 & 2 & -1 \\ 1 & -5 & 3 \\ 2 & 0 & 1 \end{bmatrix}$$

We will need to use the 3x3 identity matirx.

```
> Id3:=Matrix(3,3,shape=identity);
```

$$Id3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

We find the eigenvalues by solving the characteristic polynomial. We first find the characteristic polynomial from the definition.

```
> p:=Determinant(A-lambda*Id3);
```

$$p := -\lambda\,(5+\lambda)\,(-1+\lambda)$$

We get the opposite polynomial from CharacteristicPolynomial since it uses det(lambda $I - A$) instead of det($A -$ lambda I). However, each has the same roots or zeroes.

```
> p:=CharacteristicPolynomial(A,lambda);
```

$$p := \lambda^3 + 4\,\lambda^2 - 5\,\lambda$$

We solve the characteristic equation to find the eigenvalues, which may be real or complex.

```
> eigval:=solve(p=0,lambda);
```

$$eigval := 0,\ 1,\ -5$$

For each eigenvalue, we find corresponding eigenvectors. Note that the scalar multiple of an eigenvector is also an eigenvector corresponding to the same eigenvalue. For each eigenvalue lambda, we solve the equation (A-lambdaI)***x**=**0** to find corresponding eigenvectors.

```
> z:=Matrix(3,1,shape=zero):
  for i from 1 to 3 do
  eigvec[i]:=LinearSolve(<A-eigval[i]*Id3|z>);
  od;
```

$$eigvec_1 := \begin{bmatrix} -\_t_2 \\ \_t_2 \\ 2\,\_t_2 \end{bmatrix}$$

$$eigvec_2 := \begin{bmatrix} 0 \\ \_t0_2 \\ 2\,\_t0_2 \end{bmatrix}$$

$$eigvec_3 := \begin{bmatrix} -3\,\_t1_3 \\ 8\,\_t1_3 \\ \_t1_3 \end{bmatrix}$$

We can find specific eigenvectors by substituting any real or complex number for $\_t0_i$, for example 1.

A second way to find eigenvalues and eigenvectors is to use the commands Eigenvalues and Eigenvectors from the **LinearAlgebra** package to find the eigenvalues and eigenvectors.

```
> eigval:=Eigenvalues(A,output=list);
```

$$eigval := [0, 1, -5]$$

```
> v:=Eigenvectors(A,output=list);
```

$$v := \left[\left[0, 1, \left\{ \begin{bmatrix} -\frac{1}{2} \\ \frac{1}{2} \\ 1 \end{bmatrix} \right\} \right], \left[1, 1, \left\{ \begin{bmatrix} 0 \\ \frac{1}{2} \\ 1 \end{bmatrix} \right\} \right], \left[-5, 1, \left\{ \begin{bmatrix} -3 \\ 8 \\ 1 \end{bmatrix} \right\} \right] \right]$$

The following command picks out the second eigenvector along with its eigenvalue from the list.

```
> v[2];
```

$$\left[1, 1, \left\{ \begin{bmatrix} 0 \\ \frac{1}{2} \\ 1 \end{bmatrix} \right\} \right]$$

The first entry gives the eigenvalue corresponding to the eigenvector, and the second its multiplicity.

```
> v[2,1];
```

$$1$$

```
> v[2,2];
```

$$1$$

The third entry is a list of basis eigenvectors for the particular eigenvalue. Eigenvalues of multiplicity greater than 1 may have multiple basis eigenvectors. You can get all the eigenvectors corresponding to the particular eigenvalue by taking all linear combinations of these eigenvectors.

```
> v[2,3];
```

$$\left\{ \begin{bmatrix} 0 \\ \frac{1}{2} \\ 1 \end{bmatrix} \right\}$$

The next command picks out the first (and in this case only) basis eigenvector.

```
> v[2,3,1];
```

$$\begin{bmatrix} 0 \\ \frac{1}{2} \\ 1 \end{bmatrix}$$

Finally, we pick out the third component of this eigenvector.

```
> v[2,3,1][3];
```

$$1$$

Next, we look at the case where we have complex eigenvalues.

```
> A:=Matrix(3,3,[[1,0,2],[0,1,-1],[-1,1,1]]);
```

$$A := \begin{bmatrix} 1 & 0 & 2 \\ 0 & 1 & -1 \\ -1 & 1 & 1 \end{bmatrix}$$

```
> p:=CharacteristicPolynomial(A,lambda);
```

$$p := -4 + \lambda^3 - 3\lambda^2 + 6\lambda$$

```
> eigval:=solve(p,lambda);
```

$$eigval := 1,\ 1 + I\sqrt{3},\ 1 - I\sqrt{3}$$

```
> z:=Matrix(3,1,shape=zero):
  for i from 1 to 3 do
  eigvec[i]:=LinearSolve(<A-eigval[i]*Id3|z>);
  od;
```

$$eigvec_1 := \begin{bmatrix} \_t2_2 \\ \_t2_2 \\ 0 \end{bmatrix}$$

$$eigvec_2 := \begin{bmatrix} \_t3_1 \\ -\frac{1}{2}\,\_t3_1 \\ \frac{1}{2}\, I\sqrt{3}\,\_t3_1 \end{bmatrix}$$

$$eigvec_3 := \begin{bmatrix} \_t4_1 \\ -\frac{1}{2}\,\_t4_1 \\ -\frac{1}{2}\, I\sqrt{3}\,\_t4_1 \end{bmatrix}$$

We notice that complex eigenvalues come in conjugate pairs. If we look carefully, we can also see a relationship between the corresponding eigenvectors. Each component of corresponding eigenvectors has the same real part, but conjugate imaginary parts.

```
> eigval:=Eigenvalues(A,output=list);
```

$$eigval := \left[1, 1 + I\sqrt{3}, 1 - I\sqrt{3}\right]$$

```
> v:=Eigenvectors(A,output=list);
```

$$v := \left[\left[1 + I\sqrt{3}, 1, \left\{\left[\begin{array}{c} -\frac{2}{3} I\sqrt{3} \\ \frac{1}{3} I\sqrt{3} \\ 1 \end{array}\right]\right\}\right], \left[1 - I\sqrt{3}, 1, \left\{\left[\begin{array}{c} \frac{2}{3} I\sqrt{3} \\ -\frac{1}{3} I\sqrt{3} \\ 1 \end{array}\right]\right\}\right], \left[1, 1, \left\{\left[\begin{array}{c} 1 \\ 1 \\ 0 \end{array}\right]\right\}\right]\right]$$

We nowlook at an example of a matrix with an eigenvalue of multiplicity greater than 1.

```
> A:=Matrix(3,3,[[1,1,1],[2,2,2],[3,3,3]]);
```

$$A := \left[\begin{array}{ccc} 1 & 1 & 1 \\ 2 & 2 & 2 \\ 3 & 3 & 3 \end{array}\right]$$

```
> Eigenvalues(A,output=list);
```

$$[0, 0, 6]$$

```
> Eigenvectors(A,output=list);
```

$$\left[\left[6, 1, \left\{\left[\begin{array}{c} \frac{1}{3} \\ \frac{2}{3} \\ 1 \end{array}\right]\right\}\right], \left[0, 2, \left\{\left[\begin{array}{c} -1 \\ 0 \\ 1 \end{array}\right], \left[\begin{array}{c} -1 \\ 1 \\ 0 \end{array}\right]\right\}\right]\right]$$

Finally, we look at the relationship between the spectral radius of a matrix and its l_2 norm.

```
> A:=Matrix(3,3,[[0,2,-1],[1,-5,3],[2,0,1]]);
```

$$A := \left[\begin{array}{ccc} 0 & 2 & -1 \\ 1 & -5 & 3 \\ 2 & 0 & 1 \end{array}\right]$$

We find the l_2 norm.

```
> Norm(A,2);
```

$$\sqrt{RootOf(\_Z^2 - 45\_Z + 180, index = 2)}$$

```
> evalf(%);
```

$$6.368861112$$

Next, we find the spectral radius of A^tA and take its square root to get it's l_2 norm.

```
> M:=Multiply(Transpose(A),A);
```

$$M := \left[\begin{array}{ccc} 5 & -5 & 5 \\ -5 & 29 & -17 \\ 5 & -17 & 11 \end{array}\right]$$

```
> eigval:=Eigenvalues(M,output=list);
```

$$eigval := \left[0, \frac{45}{2} + \frac{3}{2}\sqrt{145}, \frac{45}{2} - \frac{3}{2}\sqrt{145}\right]$$

```
> rho:=max(abs(eigval[1]),abs(eigval[2]),abs(eigval[3]));
```

$$\rho := \frac{45}{2} + \frac{3}{2}\sqrt{145}$$

```
> alt2norm:=sqrt(rho);
```

$$alt2norm := \frac{1}{2}\sqrt{90 + 6\sqrt{145}}$$

```
> evalf(%);
```

$$6.368861115$$

Except for round-off error, the same as above.